# $Id: PKGBUILD 82845 2013-01-25 01:03:39Z kkeen $
# Maintainer: Kyle Keen <keenerd@gmail.com>
# Contributor: Kevin Piche <kevin@archlinux.org>
# Contributor: Roberto Carvajal <roberto@archlinux.org>

pkgname=multitail
pkgver=5.2.12
pkgrel=1
pkgdesc="Lets you view one or multiple files like the original tail program"
arch=('i686' 'x86_64')
license=('GPL')
url="http://www.vanheusden.com/multitail"
depends=('ncurses')
backup=('etc/multitail.conf')
source=("${url}/${pkgname}-${pkgver}.tgz")
md5sums=('15fc4a54778327192b0b5f1c0005a1b3')

build() {
  cd "$srcdir/$pkgname-$pkgver"
  make
}

package() {
  cd "$srcdir/$pkgname-$pkgver"
  install -D -m 755 multitail      "$pkgdir/usr/bin/multitail"
  install -D -m 644 multitail.1    "$pkgdir/usr/share/man/man1/multitail.1"
  install -D -m 644 multitail.conf "$pkgdir/etc/multitail.conf"
}
# vim: ts=2 sw=2 et ft=sh
